



Kalaschnikow S.N.:

PQLib - A Modelica Library for Power Quality analysis in Networks
2nd International Modelica Conference, Proceedings, pp. 117-122

Paper presented at the 2nd International Modelica Conference, March 18-19, 2002,
Deutsches Zentrum für Luft- und Raumfahrt e.V. (DLR), Oberpfaffenhofen, Germany.

All papers of this workshop can be downloaded from
<http://www.Modelica.org/Conference2002/papers.shtml>

Program Committee:

- Martin Otter, Deutsches Zentrum für Luft- und Raumfahrt e.V. (DLR), Institut für Robotik und Mechatronik, Oberpfaffenhofen, Germany (chairman of the program committee).
- Hilding Elmqvist, Dynasim AB, Lund, Sweden.
- Peter Fritzson, PELAB, Department of Computer and Information Science, Linköping University, Sweden.

Local organizers:

Martin Otter, Astrid Jaschinski, Christian Schweiger, Erika Woeller, Johann Bals,
Deutsches Zentrum für Luft- und Raumfahrt e.V. (DLR), Institut für Robotik und Mechatronik, Oberpfaffenhofen, Germany

PQLib- A Modelica Library for Power Quality analysis in Networks

Sergej N. Kalaschnikow

VA TECH ELIN EBG Elektronik GmbH
Ruthnergasse 1, A-1210 Vienna Austria
e-mail: s.kalaschnikow@eel.elinebg.at

Abstract

In power supply networks, the quality of the voltage is becoming more and more of a determining factor.

Non-linear loads such as diode or thyristor converters contribute to the degradation of the supply voltage quality. Non-sinusoidal currents of the non-linear loads result in the distortion of the supply voltage wave form at the point of common coupling due to the finite supply impedance.

To improve the power quality of the supply voltage active filters and reactive current compensators are used. The optimal power rating and topology of these units are very important, but also the determination of the best compensation strategy for a specific application is very important as well. Different topologies and strategies can often perform related compensation functions, resulting in a situation where financial implications determine the best solution. In this situation the power quality analysis in network using simulation tools is very useful.

This paper describes a Modelica library called PQLib (**P**ower **Q**uality **L**ibrary) designed for power quality analysis in networks using simulation tools written in Modelica.

The PQLib contains the following components:

- Definition of connectors for three phase networks
- Models for:
 - three phase passive electrical elements like resistor, capacitor and so on.
 - three phase electrical machines and transformers
 - three phase transmission lines
 - semiconductor controlled dc and ac electrical drives

- power factor correction devices (passive filters)
- mains active restoring devices using semiconductors (active filters)
- measuring instruments: true rms voltmeter and amperemeter, digital frequency analyser
- Examples

1. Building of the PQLib

The PQLib is based on the package concept. The package concept was introduced into Modelica to help organize definitions of models, connectors, etc. [1, 2]. Fig.1 shows the components of the package PQLib.

1.1. Types

In the PQLib for currents, voltages and impedances the per unit (p.u.) quantities with the definitions according to [3] are used. Currents and voltages are related to their rated peak- values:



Fig. 1. Components of the package PQLib

$$u(t) = \frac{U(t)}{\sqrt{2} \cdot U_N}, \quad i(t) = \frac{I(t)}{\sqrt{2} \cdot I_N}. \quad (1)$$

Impedances are referred in the same way as (1) to

$$z(\omega) = \frac{Z(\omega) \cdot I_N}{U_N}, \quad (2)$$

with: U_N and I_N are nominal values of the voltage and the current accordingly.

Consequently, the types in the PQLib are defined as (for example for the first phase):

```

type Voltage1stPhase = Real (
  final quantity="Voltage",
  final unit="p.u.",
  displayUnit="p.u.");
type Current1stPhase = Real (
  final quantity="Current",
  final unit="p.u.",
  displayUnit="p.u.");
type Resistance = Real (
  final quantity="Resistance",
  final unit="p.u.",
  min=0,
  displayUnit="p.u.");
type Conductance = Real (
  final quantity="Conductance",
  final unit="p.u.",
  min=0,
  displayUnit="p.u.");
type Reactance = Real (
  final quantity="Reactance",
  final unit="p.u.",
  min=0,
  displayUnit="p.u.");
type SignalAnalog = Real;
type SignalBoolean = Boolean;
type SignalDiscrete = Real;

```

1.2. Interfaces

Usually every package includes some interface definitions which are used throughout the package. In the PQLib package the basic interface definition is the three phase pin, which is a connector. At the pin the pin three phase voltages va, vb and vc and the pin three phase currents ia, ib and ic are defined. The positive pin is described in the following way:

```

connector Pin3Ph
  Voltage1stPhase va;
  Voltage2ndPhase vb;
  Voltage3rdPhase vc;
  flow Current1stPhase ia;
  flow Current2ndPhase ib;
  flow Current3rdPhase ic;
end Pin3Ph

```

The negative pin differs in its graphical representation only.

The TwoPin interface is defined as a partial model:

```

partial model TwoPin3Ph
  PQLib.Interfaces.Voltage1stPhase Vr;
  PQLib.Interfaces.Voltage2ndPhase Vs;
  PQLib.Interfaces.Voltage3rdPhase Vt;

```

```

  PQLib.Interfaces.Current1stPhase Ir;
  PQLib.Interfaces.Current2ndPhase Is;
  PQLib.Interfaces.Current3rdPhase It;

```

```

  PQLib.Interfaces.Pin3Ph P;
  PQLib.Interfaces.NegPin3Ph N;

```

```

equation
  Vr = P.va - N.va;
  Vs = P.vb - N.vb;
  Vt = P.vc - N.vc;
  P.ia + N.ia = 0;
  P.ib + N.ib = 0;
  P.ic + N.ic = 0;
  Ir = P.ia;
  Is = P.ib;
  It = P.ic;
end TwoPin3Ph

```

For the control package of the PQLib the analog and digital as well as logical interfaces are defined in the classical way of the Modelica interface definition with the exception of the definition for three phase vectors [3]. For example, the vector of voltages \mathbf{u} is derived from the instantaneous values of the three phase voltages u_a , u_b and u_c as follows:

$$\mathbf{u} = \frac{2}{3} \left(u_a + u_b \cdot e^{j\frac{2\pi}{3}} + u_c \cdot e^{-j\frac{2\pi}{3}} \right) = u_\alpha + ju_\beta \quad (3)$$

Thus, the connectors for three phase vectors can be described in the following way:

```

connector InAB
  input SignalAnalog alfa;
  input SignalAnalog beta;
end InAB

```

```

connector OutAB
  output SignalAnalog alfa;
  output SignalAnalog beta;
end OutAB

```

2. Main Components of the PQLib

Any network consists of passive electrical elements like resistors, capacitors and so on. The three phase transmission line itself can be

represented as a circuit of passive electrical elements. The passive shunt harmonic filter, which is the traditional method of controlling harmonic distortion levels, consists of a tuning reactor in series with a capacitor bank.

At the same time, each network consists of active electrical elements as well. These elements are: generators, electrical motors, four quadrant electrical drives, active harmonic filters and so on. The PQLib packages imply both passive and active electrical elements. Fig. 2 shows, for example, the package of electrical elements which are based on the TwoPin interface.

2.1 Passive electrical elements

The three-phase elements like resistors, capacitors and inductors are equally defined. The three phase capacitor, for example, is defined as:

```
class C
  extends PQLib.Interfaces.TwoPin3Ph;
  parameter PQLib.Interfaces.Reactance xc[3]={1,1,1};
equation
  1/w/xc[1]*der(Vr) = Ir;
  1/w/xc[2]*der(Vs) = Is;
  1/w/xc[3]*der(Vt) = It;
end C
```




Fig. 2. Components of the package Elements

In the same way the three phase switch can be defined:

```
class SW
  extends PQLib.Interfaces.TwoPin3Ph;
  parameter Real OnTime(unit="[s]") = 0 "switch ON Time";
  parameter Real Ron(final min=0) = 1.E-5 "Closed switch resistance";
  parameter Real Goff=1.E-5 "Opened switch conductance";
protected
  Real s1;
  Real s2;
  Real s3;
equation
  Vr = s1*(if time >= OnTime then Ron else 1);
  Ir = s1*(if time >= OnTime then 1 else Goff);
  Vs = s2*(if time >= OnTime then Ron else 1);
  Is = s2*(if time >= OnTime then 1 else Goff);
  Vt = s3*(if time >= OnTime then Ron else 1);
  It = s3*(if time >= OnTime then 1 else Goff);
end SW
```

To get the star connection of the three phase elements the class Y can be used:

```
class Y
  PQLib.Interfaces.Pin3Ph v0;
equation
  v0.ia + v0.ib + v0.ic = 0;
  v0.va = v0.vb;
  v0.vb = v0.vc;
end Y
```

To use one-phase electrical elements of the Modelica standard library the class Connector3Ph (see Fig. 2 and Fig.3) is used. The class Connector3Ph is described in the following way:

```
class Connector3Ph
  PQLib.Interfaces.Pin3Ph InOut3Ph;
  Modelica.Electrical.Analog.Interfaces.Pin Ph1;
  Modelica.Electrical.Analog.Interfaces.Pin Ph2;
  Modelica.Electrical.Analog.Interfaces.Pin Ph3;
equation
  InOut3Ph.va = Ph1.v;
  InOut3Ph.vb = Ph2.v;
  InOut3Ph.vc = Ph3.v;
  InOut3Ph.ia = -Ph1.i;
  InOut3Ph.ib = -Ph2.i;
  InOut3Ph.ic = -Ph3.i;
end Connector3Ph
```

The other passive elements like the three-phase full wave converter, the three-phase transformers, the passive harmonic filters and so on are created by using graphical model editing tools. Fig. 3 shows, for example, a model of the three-phase full wave converter.

2.2 Active electrical elements

The voltage source is defined in the following way:




Fig.3. Diagram of the three-phase full wave converter

```

class VS
  extends PQLib.Interfaces.TwoPin3Ph;
  parameter Real N_harmonic[:]={0,0} "Array of
  numbers of harmonics";
  parameter Amplitude
  V_harmonic[size(N_harmonic, 1)]= {0,0};
  parameter Phase Ph_harmonic[size(N_harmonic,
  1)]= {0,0};
  parameter Amplitude V1=1.0;
  parameter Phase Ph1=0;
  parameter Amplitude V1_opposite=0.0;
  parameter Phase Ph1_opposite=0;

equation

  Vr = V1*cos(w*time + Ph1*pi/180) +
  V1_opposite*cos(w*time + Ph1_opposite*pi/
  180) + V_harmonic*cos(N_harmonic*w*time +
  Ph_harmonic*pi/180);

  Vs = V1*cos(w*time - 2*pi/3 + Ph1*pi/180) +
  V1_opposite*cos(w*time + 2*pi/3
  + Ph1_opposite*pi/180) +
  V_harmonic*cos(N_harmonic*(w*time - 2*pi/3) +
  Ph_harmonic*pi/180);

  Vt = V1*cos(w*time + 2*pi/3 + Ph1*pi/180) +
  V1_opposite*cos(w*time - 2*pi/3
  + Ph1_opposite*pi/180) +
  V_harmonic*cos(N_harmonic*(w*time + 2*pi/3) +
  Ph_harmonic*pi/180);

end VS
    
```

This voltage course definition makes it possible to simulate all possible kinds of the voltage distortion in industrial supply systems.

The three-phase current source is defined in the same way.

The very important part of the PQLib package are the switch mode power devices such as four quadrant frequency controlled electrical drives and active harmonic filters. These devices use the pulse-width modulation IGBT-inverter technology. Fig. 4 shows basic configuration of a IGBT-inverter .




Fig. 4. Basic configuration of a IGBT-inverter

Owing to the fact that the goal of the power quality analysis is to study the network itself and in order to simplify the model of the IGBT-inverter, the following equivalent circuit for IGBT-inverters can be used [3]:




Fig. 5. Equivalent circuit for IGBT-inverters

Assuming a balanced three-phase system without the neutral connection and neglecting the resistance of the power switches, the circuit in

Fig.5 with a voltage-source inverter can be described as [3]:

$$\left. \begin{aligned} L \frac{d\mathbf{i}_F}{dt} &= \mathbf{u}_S - \mathbf{u}_W - R\mathbf{i}_F \\ C \frac{du_d}{dt} &= \frac{3}{2} \sigma(n) \operatorname{Re}(\mathbf{i}_F e^{-j\frac{\pi}{3}(n-1)}) - i_L \end{aligned} \right\} n = 0,1,2,\dots,6 \quad (1)$$

with

$$\mathbf{u}_W = \frac{2}{3} \sigma(n) u_d e^{j\frac{\pi}{3}(n-1)},$$

where the used symbols denote:

- \mathbf{i}_F complex vector of the line currents;
- \mathbf{u}_S complex vector of the mains voltage;
- \mathbf{u}_W complex vector of the inverter voltage;
- u_d dc-link voltage;
- i_d dc-link current;
- i_L dc-link load current;
- L inductance of the line choke;
- R resistance of the line choke;
- C capacitance of the dc-link capacitor;
- n switching- state of the converter (Fig.5)
- σ switching function: $\sigma(n) = \begin{cases} 1, & \text{if } n > 0 \\ 0, & \text{if } n = 0 \end{cases}$.

For more information about the control unit of IGBT- inverters see [4]. The description of the control system for the active filter for industrial mains can be found, for example, in [5]. The control system of the four quadrant adjustable speed drives is described in [6].

2.3 Measuring instruments

Often the goal of the power quality analysis in networks is to get a value of the total harmonic distortion factor (THD) or values of harmonic amplitudes at the point of common coupling (PCC). In this case, it is very useful to use the built-in measuring instruments.

Fig. 6 shows the components of the package Instruments. There are instruments for voltages and currents. Together with model RMS it is possible to measure the rms-values of the three-phase voltages and currents and the THD-factor as




Fig.6. Components of the package Instruments

well. To get the spectra for voltages and currents the class DFT can be used. The class DFTG is very useful for the measuring of single reference number harmonics. The classes DFT and DFTG are based on DFT-technique .

Several aspects of the PQLib-package usage are demonstrated in the following example.

3. Examples

Non-linear loads such as diode or thyristor converters contribute to the degradation of the supply quality. Non-sinusoidal currents of the non-linear loads result in the distortion of the supply voltage wave form at the point of common coupling due to the finite supply impedance.

In industrial mains, the passive filters have traditionally been used to absorb harmonics generated by the load, primarily due to their low cost and high efficiency. This is a good approach when power factor correction is needed too. However, they have the following drawbacks:

- the mains impedance strongly influences the compensation characteristics of the filter;

- they result in new resonances and therefore magnify the levels of the other harmonics;

Compared with the passive filter, the active filters can be used to reduce harmonics in the industrial mains without worrying about all the problems associated with applying passive filters. Additionally they can not be overloaded by harmonics from the power system. Due to the fact that active filters use the same IGBT-inverter technology that is used in adjustable speed drives, their cost is not high.

The next Modelica model shows the utilization of the active filter to achieve harmonic cancellation for an adjustable speed drive (see Fig.7). The simulation results are shown in Fig. 8-10.




Fig. 7. Using active filter connection to achieve harmonic cancellation for adjustable speed drives




Fig. 8. Simulation results: one-phase current of the adjustable speed drive

The current of the adjustable speed drive in one phase is shown in Fig.8. Fig. 9 shows the current of the active filter in the same phase.




Fig. 9. Simulation results: one-phase current of the active filter




Fig. 10. Simulation results: one-phase current in the mains

The current in the mains (sum current) is presented in Fig.10. From Fig.10 it is seen that the sum current has practically sinusoidal wave form. The harmonics of the ac drive current are practically eliminated.

4. Conclusion

The presented Modelica package PQLib is very useful for power quality analysis in networks. Using the library, the user can quickly create the network with different kind of mains loads and measuring instruments using the graphical editor of the Dymola. The library has an open structure and all models can be modified.

References

- [1] Tiller, M.: Introduction to physical modeling with Modelica, Kluwer Academic Publishers, 2001
- [2] Otter, M.; Elmqvist, H.; Mattsson, S.E.: Objektorientierte Modellierung physikalischer Systeme, Teil 1-8. at Automatisierungstechnik, 1999.
- [3] Kalachnikov, S.: Regelung des netzseitigen Puls-stromrichters eines Vier-Quadranten-Spannungszwischenkreis-Umrichters, ELIN-Zeitschrift, Heft 3/4 pp. 46-55, 1994 (in German)
- [4] Mohan, N., Undeland T.M. und Robbins, W.P. Power Electronics, John Wiley & Sons, Inc. 1995, pp.805
- [5] Kalachnikov S., Berger, H.: AC-Drive with Three-Phase PWM-Rectifier as a Reactive Power Compensator, IEEE Stockholm Power Tech Conference, Stockholm, Sweden, June 18-22, 1995, Electrical Machines and Drives, pp 426-431
- [6] Kalachnikov, S., Wieser, R.: AC Drives with IGBT Recovery Inverter System in the Power Range up to 500 kW, PEMC '96, Budapest, Hungary, September, 1996, pp 3.151-3.155